
*
*                        ^  VCC                           ^ VCC
*                       /_\                              /_\
*                       _|_                              _|_
*                      / _ \                            / _ \
*                      \/ \/ ITAIL                      \/ \/  I2STAGE
*                      /\_/\                            /\_/\
*                      \___/                            \___/
*                        |  VTAIL                         |
*                     ___|___                             |
*                    |       |                ____________|
*           VP2B   <-         ->    VP3B    _|_           |
*               _|' QP2      QP3 `|_       |OUT|          |
*              | |`_           _ '| |      |___|          |
*     ___    <-     |         |      ->     _|_           |
*    |INP|_|' QP1   |         |     QP4 `|_|INN|    CCOMP |
*    |___| |`_      |VN3B     |       _ '| |___|    __||__|
*     _|__    _|_   |________/|\_   _|_            |  ||  |
*    /_   \   ///   |    _____|  |  ///            |     _|
*   // \   \        |_   |   _|  |_________________|___|'QN3
*   \   \_//     QN1  `|_|_|' QN2                      |`->
*    \____/         <-'|   |`->                          _|_
*      |           _|_  VN2B _|_                         ///
*     _|_          ///       ///                   SECOND STAGE
*     ///        TURN AROUND STAGE
*                
*
.OPTIONS  method=gear  GMIN=1e-18

VIN       INP    0      SIN(6   6.7      1K )   
QP1       0      INP    VP2B    PNPL   1
QP2       VN3B   VP2B   VTAIL   PNPL   1
QP3       VN2B   VP3B   VTAIL   PNPL   1
QP4       0      OUT    VP3B    PNPL   1

QN1       VN3B   VN2B   0       NPNV   1
QN2       VN2B   VN2B   0       NPNV   1
QN3       OUT    VN3B   0       NPNV   1

VCC       VCC    0      12   
ITAIL     VCC    VTAIL  50u
I2STAGE   VCC    OUT    100u  

CCOMP     OUT    VN3B   20p

.tran     1u     2m     0      1u     

.model      NPNV npn BF=150

.model      PNPL pnp BF=5



.control
run
set pensize = 2
plot v(inp)  v(out) 

.endc

.end

* ====================END================================

The LM324 is perhaps the best example of a product designed for
the automotive market. At the time there was only a single 12volt
supply available and hence the development of single supply Op Amps.

Its not that single supply Op Amps can't be used on Dual supplies or
visa versa. The single supply title assumes certain input and output
features.

For instance the LM324 input can function with both it inputs at the
ground level on a single 12V supply. The LM741's inputs cannot
do this. The automotive amplifiers were design to operate at a much
lower supply voltage. In the case of the LM324, the output can swing
to a saturation voltage above ground to give more output voltage swing. 



This simulation shows some of the extra features that get designed
within an Op Amp that may not appear in the data sheet. If care is
not taken at the input stage, swinging a few hundred millivolts below
ground can cause the output voltage swing to reverse phase. 

*                      /\_/\                            /\_/\
*                      \___/                            \___/
*                        |  VTAIL                         |

*           VP2B   <-         ->    VP3B    _|_           |

*     _|_          ///       ///                   SECOND STAGE

The circuit is a simplified LM324 input stage. As long as lowering the 
base voltage on QP2 increases the base voltage on QN3, then the 
amplifier is operating normally. If the base of QP2 drops enough
to saturate QP2, this is no longer the case.

It had become somewhat of a design standard that the input stage should
be able to operate 200mV below the supply voltage over full temperature.
The automotive customers seem happy with this level. When RRIO amplifiers
came into existence, this spec carried over to being able to swing the 
inputs 200mV beyond both supplies. 


